
Initially I was using ‘/sharedStates/’ variables to save the values of variables where the scope of the variables spanned would be a given session, but was not really sure about the scoping or lifetime of these variables.

Upon learning that these variables are passed between the client and the host server each request/response cycle, I changed my approach to saving these values, upon suggestion from Lansa, to storing the values in a table on the host server. The table, ‘LRSTATE’, simply stores these values as key/value pairs, each associated with a key that represents the client.

The variables are stored using procedure ‘SetStateValue()’, which takes as parameters the client GUID (provided in the ‘systemINFO’ data structure), the ‘key’, and the ‘value’.

Procedure ‘GetStateValue()’ returns the ‘value’ associated with the client GUID and the ‘key’ passed in.

The data in table ‘LRSTATE’ looks like:

The code for the ‘GetStateValue()’ and ‘SetStateValue()’ procedures, the DDL for table ‘LRSTATE’, and some sample calls to the procedures are shown below.

Feel free to use any of this that might be of help to your situation
DDL for table ‘LRSTATES’:
--------------------------------------------------------------------------------
-- Table        - LRSTATE
-- Description  - Long Range Mobile Application State Variables
-- Author       - John Joiner
-- Date Created - 03/06/2013
--------------------------------------------------------------------------------
-- Purpose:
--
--     This table holds information pertaining to mobile applications written
--     with Lansa LongRange that should persist between sessions.

--     While LongRange's '/sharedState' variables persist between sessions, these
--     variables must be passed between client and server with each request and
--     response. Too many of these '/sharedState' variables can degrade
--     performance.
--
--------------------------------------------------------------------------------

create table LRSTATE
   (chgZ       generated always for each row on update as row change timestamp
                  not null,
    chgUsr     char    (18) ccsid 37 not null with default USER,
    clientID   char    (40) ccsid 37 not null,
    key        varchar (256) allocate(50) ccsid 37 not null,
    value      varchar (256) allocate(50) ccsid 37 not null,

       constraint LRSTATE_PK_CLIENTID_KEY primary key (clientID, key))

       rcdFmt LRSTATER;

label on table LRSTATE is 'Long Range Mobile Application State Variables';

label on column LRSTATE
              --    ...+... 1 ...+... 2 ...+... 3 ...+... 4 ...+... 5 ...+... 6
   (chgZ       is 'Change              Timestamp',
    chgUsr     is 'Change              User',
    clientID   is 'LongRange           Client              ID',
    key        is 'State               Variable            Key',
    value      is 'State               Variable            Value');

label on column LRSTATE
   (chgZ       text is 'Change Timestamp',
    chgUsr     text is 'Change User',
    clientID   text is 'LongRange Client ID',
    key        text is 'State Variable Key',
    value      text is 'State Variable Value');

GetStateValue Procedure:
      // =======================================================================
      // Function GetStateValue:
      // =======================================================================
      // This function retrieves the value of the state variable for the client
      // ID passed.
      // =======================================================================
     P GetStateValue   B                   export
     D                 PI                  like(tLRStateVarsDS.value)
     D  inClientID                         like(tLRStateVarsDS.clientID) const
     D  inKey                              like(tLRStateVarsDS.key) const

      // =======================================================================
      //- Local Constants -
      // =======================================================================

      // =======================================================================
      //- Local Variables -
      // =======================================================================

     D value           S                   like(tLRStateVarsDS.value) inz

       // Copy Member SQLCMPOPTN: Compiler Options for Applications with Embedded
       //                         SQL
      /copy *libl/COPYBOOKS,SQLCMPOPTN

      /free

       // override defaults from SQLCMPOPTN
       exec sql set option closqlcsr = *endmod, commit = *none;

       // Client ID and Key required
       if inClientID = *blanks
       or inKey = *blanks;
          return *blanks;
       endif;

       // Retrieve value of state variable
       exec sql select lrState.value
                into   :value
                from   lrState
                where      clientID = :inClientID
                       and key = :inKey;

       return value;

      /end-free

     P GetStateValue   E
SetStateValue Procedure:
      // =======================================================================
      // Function SetStateValue:
      // =======================================================================
      // This function adds/updates the value of the state variable for the client
      // ID passed.
      // =======================================================================
     P SetStateValue   B                   export
     D                 PI
     D  inClientID                         like(tLRStateVarsDS.clientID) const
     D  inKey                              like(tLRStateVarsDS.key) const
     D  inValue                            like(tLRStateVarsDS.value) const

      // =======================================================================
      //?- Local Constants -
      // =======================================================================

      // =======================================================================
      //?- Local Variables -
      // =======================================================================

     D rowFnd          S               N   inz(FALSE)

      /free

       // Client ID and Key required
       if inClientID = *blanks
       or inKey = *blanks;
          return;
       endif;

       // If 'Value' is blank, delete existing state variable
       if inValue = *blanks;
          exec sql delete from lrState
                   where           lrState.clientID = :inClientID
                               and lrState.key = :inKey;
          return;
       endif;

       // Determine if state variable already exists
       exec sql select '1'
                into   :rowFnd
                from   lrState
                where      clientID = :inClientID
                       and key = :inKey;

       // Update existing
       if rowFnd;
          exec sql update lrState
                   set    (chgUsr, value) = (default, :inValue)
                   where      clientID = :inClientID
                          and key = :inKey;

       // Add state variable
       else;
          exec sql insert into lrState (clientID, key, value)
                   values      (:inClientID, :inKey, :inValue);
       endif;

       return;

      /end-free

     P SetStateValue   E

Prototypes and templates:
      /if defined(GetStateValue)
      // =======================================================================
      // Function GetStateValue:
      // =======================================================================
      // This function retrieves the value of the state variable for the client
      // ID passed.
      // =======================================================================
     D GetStateValue   PR                  extproc('GetStateValue')
     D                                     like(tLRStateVarsDS.value)
     D  INPUT_Long_Range_Client_ID...
     D                                     like(tLRStateVarsDS.clientID) const
     D  INPUT_State_Variable_Key...
     D                                     like(tLRStateVarsDS.key) const
      /undefine GetStateValue
      /endif

      /if defined(SetStateValue)
      // =======================================================================
      // Function SetStateValue:
      // =======================================================================
      // This function adds/updates the value of the state variable for the client
      // ID passed.
      // =======================================================================
     D SetStateValue   PR                  extproc('SetStateValue')
     D  INPUT_Long_Range_Client_ID...
     D                                     like(tLRStateVarsDS.clientID) const
     D  INPUT_State_Variable_Key...
     D                                     like(tLRStateVarsDS.key) const
     D  INPUT_State_Variable_Value...
     D                                     like(tLRStateVarsDS.value) const
      /undefine SetStateValue
      /endif

      /if defined(tDfn_LRCOM)
     D tLRStateVarsDS...
     D                 DS                  template qualified
     D  clientID                    256A   varying inz
     D  key                         256A   varying inz
     D  value                       256A   varying inz

     D tLRNG_RqstAct   S             32A   template
     D tLRNG_RqstPgm   S             10A   template

     D tLRNG_User...
     D                 S             10A   template inz
     D tLRNG_Password...
     D                 S            128A   template inz
      /undefine tDfn_LRCOM
      /endif

Example call to ‘GetStateValue()’:
       wkTripA = GetStateValue(inSystemInfo.ClientGUID :
                               'CURRENT_ID1_TRIP');

Example call to ‘SetStateValue()’:
       if not geoLocationOn;
          LRNG_AssignStrToProp('default' : '/system.geo.updateFrequency');
          SetStateValue(inSystemInfo.ClientGUID : 'geoLocationOn' : '1');
       endif;

